
Introduction

Customers are looking for products on the market faster than ever before. Customers are also demanding the products work right the first time.  In an effort to address customer demands many organizations are adopting agile techniques as a means to speed up the time to market for their products and also increase the quality of those products.  Databases are the cornerstone of many products, finding ways to speed up database development and the quality of database code is of strategic importance to many organizations.

This paper will provide real-world best practices that have helped improve the quality of database installations and code by over 80% while reducing planned maintenance windows for deployments.
What is Agility   

The most basic definition of agility can be summed up as “A rapid change of velocity or direction in response to a stimulus”.   In the software engineering field this term is used very frequently and can mean very different things depending on the audience.  Many terms around agile development exist such as Test Driven Development, Scrum, Pair Programming, or Extreme Programming.  While many articles focus on the ability to rapidly change direction during the course of a project, the testing required to have the confidence to change direction quickly without fear is the true power of agile development practices.
An interesting side effect of adding the automated testing required for agility is that is can also greatly increase product quality. 
The chart in figure 1 shows the difference in the man hours required for a project using agile techniques such as automated unit testing versus a project using traditional development techniques such as manual unit testing.  The peaks in figure 1 represent code drops that are delivered to a quality assurance group.   During a typical project the automated testing will require more resources for the test creation and execution during the initial code drop.  After the initial code drop the man hours required to support defects is drastically reduces.  In most cases if no additional functionality is requests only a small code drop towards the end of the project for minor items is required as opposed to the multiple code drops to address major defects found during the test phase of a project with no automated tests.  In the absence of automated testing many more man hours are required because the same code must be reworked multiple times, manually unit tested, packaged and deployed multiple times during the testing phase of the project.   By using automated testing, multiple builds for the same component are not needed saving a large number of man hours from the project.  This is of course dependent on the quality of the tests created by the development team.  In my own experience our team was able to create an almost defect free initial code drop and did not have to re-drop any major component during the lifecycle of a major system upgrade.
Why is database testing important 

Traditional development and quality assurance roles in a software development organization live out their existence playing a form of “cops and robbers”.  In the case of software development the “robber” is the developer that commits a crime by introducing a defect into the system.  The “cop” is the quality assurance engineer who’s job it is to find the defect.  The flaw in this model is that if the quality assurance engineer cannot find the defect it is propagated to production where it causes the software to not function properly.  

This model has to be changed so the code is considered incorrect until it is can be proven by the developer based on tests to be correct.
In the “cops and robbers” model or the “guilty until proven innocent” model testing still occurs.  The only difference is the timing and quality of the testing. 

Defects will be found, the only question is when:
They can be found

· During unit/integration testing

· Defects can be found during development; at this point there is time to refactor the code ensuring it is robust and maintainable.

· During QA testing

· Quality Assurance can find the defects and the same code must be modified a second and maybe third time requiring work on both the currently scheduled tasks and defects.

· In Production
· Customers can find the defects in production,  all other work and the engineer will have to perform three days of coding overnight and then spend the rest of the week catching up on scheduled tasks and deadlines.
As stated before it is only a matter of time before the defect is found.  Finding the defects as early as possible allows time to fix the code right the first time.
PL/SQL Unit Testing 

The key to increasing the quality of software is rigorous testing.  The engineer developing the component is the most familiar with the code and can write the most exhaustive test cases.  There are many ways to test packages, when starting with no tests the task can seem very challenging.  The best way to quickly begin unit testing your existing packages is to perform a 10046 trace of database sessions that are executing the pl/sql code.  Based on the trace files test packages can be created of pl/sql blocks that calls packages in the same order as the application.  This helps to better replicate the real world use of the package and helps people not familiar with the system understand the type of data that is passed into the packages.  
Once the basic tests have been created for existing packages a testing tool such as utplsql can be incorporated to take the tests to the next level and provide even more functionality and test coverage.  

UTPLSQL

Setting up and using utplsql is very straight forward, to get started:

· Download zip file from 

· http://utplsql.sourceforge.net/
· download and Unzip the file.

· Create a user

· connect to the database as the system user
· create a user named utplsql 
· grant create session, create table, create procedure, create sequence, create view, create public synonym, drop public synonym to utplsql.
· In the code directory where the file is unzipped, login to sqlplus as utplsql and run @ut_i_do install (you must pass in the parameter “install”).

· Create public synonyms for all of the packages owned by utplsql

A test package is created for each application package.  For example, to test PKGORDERITEMS a package called UT_ PKGORDERITEMS is created.  This allows for automation.

Each package contains a ut_setup, ut_teardown, and then individual procedures to test the code.  The sequence of execution is based on alphabetical order.

Tests are based on assertions.  The tests assert the result you expect to be returned from the procedure.  Assertions are the heart of the testing framework.  In the testing package the target procedure is called and then an assertion to verify it worked as expected is called.  The assertions will prove that the results are correct, instead of just hoping the results are correct.  

The tests are self-contained, which means no parameters can be passed into the unit tests.  Everything must be specified in the unit test itself or in the ut_setup procedure. 
Below are the basic assertions that can satisfy 80% of test cases.

· utassert.eqquery – compares multiple columns of two queries or result sets.

· utassert.eqqueryvalue – compares a single column returned from a query.

· utassert.eq –compares two values.

· utassert.throws – is used to test that exceptions are being thrown and caught properly.

For the other 20% you can also

· Compare tables, pipes, refcursors, and collections for equality.

· Check for null values, table counts and whether previous assertions passed or failed.

· You can also extend the assert procedure through inheritance and create your own.

The code below is a basic test package used to test the adding records to an OrderItems table using a package called PkgOrderItems.  The table definition can be found in Appendix A and the package can be found in Appendix B at the end of this document.
CREATE OR REPLACE PACKAGE BODY ut_PkgOrderItems AS

   PROCEDURE ut_setup

   IS

   BEGIN

   null;

   end ut_setup;

   PROCEDURE ut_teardown

      IS

   BEGIN

   null;

   end ut_teardown;

   PROCEDURE ut_1_add_order_items_proc

   IS

    vOrder_Id OrderItems.Order_Id%TYPE := 1;

    vLine_Item_Id OrderItems.Line_Item_Id%TYPE := 1;

    vProduct_Id OrderItems.Product_Id%TYPE := 1;

    vUnit_price OrderItems.Unit_Price%TYPE := 1;

    vQuantity  OrderItems.Quantity%TYPE := 2;

   BEGIN

   PkgOrderItems.Add_OrderItems(vOrder_Id,vLine_Item_Id,vProduct_Id,vUnit_price,vQuantity);

utassert.eqqueryvalue('ut_1_add_order_items_proc', 

    'select count(1) from OrderItems where Order_Id='||to_char(vOrder_id),

    1);

   end ut_1_add_order_items_proc;

END ut_PkgOrderItems;

The call to  PkgOrderItems.Add_OrderItems is required to be written by any developer in the form of a manual unit test or an automated test before the code is delivered to production.  The call to utassert.eqqueryvalue is the extra work involved using the utplsql package.  This work is worth the extra effort since the test can be run any time the package changes and it proves that it is working correctly.  Without this proof an engineer would have to manually perform the same checks every time the package changed.  The manual checks are typically done haphazardly and allow errors to propagate to production.     

To run the test the procedure utplsql.test must be called with the package name. This is one of many ways to execute the test.
SET SERVEROUTPUT ON size 1000000

BEGIN

   utplsql.test(package_in=>'PKGORDERITEMS ', recompile_in=>FALSE);  

  DBMS_OUTPUT.put_line (utplsql2.runnum);

END;

The output of the procedure is:
>    SSSS   U     U   CCC     CCC   EEEEEEE   SSSS     SSSS

>   S    S  U     U  C   C   C   C  E        S    S   S    S

>  S        U     U C     C C     C E       S        S

>   S       U     U C       C       E        S        S

>    SSSS   U     U C       C       EEEE      SSSS     SSSS

>        S  U     U C       C       E             S        S

>         S U     U C     C C     C E              S        S

>   S    S   U   U   C   C   C   C  E        S    S   S    S

>    SSSS     UUU     CCC     CCC   EEEEEEE   SSSS     SSSS

                                    OR

>  FFFFFFF   AA     III  L      U     U RRRRR   EEEEEEE

>  F        A  A     I   L      U     U R    R  E

>  F       A    A    I   L      U     U R     R E

>  F      A      A   I   L      U     U R     R E

>  FFFF   A      A   I   L      U     U RRRRRR  EEEE

>  F      AAAAAAAA   I   L      U     U R   R   E

>  F      A      A   I   L      U     U R    R  E

>  F      A      A   I   L       U   U  R     R E

>  F      A      A  III  LLLLLLL  UUU   R     R EEEEEEE

This testing provides a powerful mechanism that greatly improves the amount of code that is tested and the repeatability of those tests.
Quest Code Tester
This tool is the next generation of utplsql.  It has a GUI interface that helps the developer build and run the tests from his/her desktop.
Database Install Validation

Database schema or application installations are one-time tasks where quality is extremely critical.  Errors during a database schema install can cause entire releases to be rolled back or cause service level agreements (SLA’s) to be missed.  To help ensure the success of deployments Post and Pre Validation scripts can be used.

Validation scripts evolved based on the following scenarios that caused deployments to fail or go very badly
· Errors being written to files and never reviewed.

· Errors never being written to files at all.

· Scripts not executing, which means no error is generated.

· Scripts that exit part of the way through and never execute the remaining commands.  

When any of the errors above occurred what happened next is someone attempted to start the application and either it started and then generated cryptic errors or it did not start at all.

When maintenance windows extend for hours, engineers have the luxury of diagnosing and resolving the issue.  When the maintenance windows are measured in minutes the only options available is to miss the maintenance window or rollback the release.
By validating database installs you pinpoint errors in the installation almost immediately.  By providing very rapid diagnostic data, engineers can correct the issue during the deployment or quickly uninstall and re-install the software.

The pre-validation script below ensures that the DATA and INDEX tablespaces are present in the database before the table and indexes are created.  The PkgValidation.assert procedure is included in Appendix C and allows the user to assert that the expected value matches the actual value.  If there is not a match the procedure returns an error.

set serveroutput on size 1000000;

spool pre_validation.lst;

declare

v_actual 
varchar2(30);

begin

/***************************************

Verifies DATA Tablespace exsits.
****************************************/
select count(1) into v_actual from dba_tablespaces

  where tablespace_name = 'DATA';

PkgValidation.assert(1, v_actual,' Verify DATA TABLESPACE');

/***************************************

Verifies INDEX Tablespace exists.

****************************************/
select count(1) into v_actual from dba_tablespaces

  where tablespace_name = 'INDEX';

PkgValidation.assert(1, v_actual,' Verify INDEX TABLESPACE');

--

--

--  Print Validation Summary

--

dbms_output.put_line('=======================================================');

dbms_output.put_line('**Assertions passed : '||PkgValidation.v_assertions_passed);

dbms_output.put_line('**Assertions Failed : '||PkgValidation.v_assertions_failed);

dbms_output.put_line('**Total Assertions  : '||PkgValidation.v_total_assertions);

dbms_output.put_line('=======================================================');

end;

/

spool off;

Automated Report Testing

Reports are often the most overlooked area of a testing strategy and typically occur at the end of a project.  Reports also provide a view of the data entered by the application and show errors in application logic.  This is why they are one of the most important areas that can benefit from automated testing.   The reports themselves come in many shapes and sizes, some are comma separated text files, other are rich graphical web layouts.  If the reports are delimited text files a simple and effective method is to:

1.) Create an external table based on the file format.
2.) Create automated tests using utplsql or a stand alone package that loops through the records in the external table and compares the report data with the database.

Things you can catch:

a. Are natural joins omitting records that should exist in the report?

b. Can the sum of transaction in the database match the sum of records in the report?

c. Are the begin and end dates of the reports inclusive or exclusive of the times?

d. Can similar summary reports be reconciled with detailed line item reports?

e. Are changes in the database schema breaking any reports?

f. Are the records added as part of application unit test reflected correctly in the report?

If the reports are graphical another method is to trace the sql generated by the reports and create a test driver the runs the report sql outside of the reporting engine.   The resulting files can then be loaded in the same way described above and compared.

Creating Install, uninstall and rollback procedures 

Creating install, uninstall and rollback scripts is vital for any database deployment.  Many times developers create elaborate frameworks that are more complicated than the installation process itself.  The most simple and effective build strategy is simply using sqlplus and standard naming conventions.  

Installs

The most effective install scripts are ones that are very intuitive.

Below is an example of an effective strategy.
Create a base directory to copy all of the sql files into such as /orders/release1.0/install

1.) Call a pre_validation script to verify pre-conditions such as the tablespaces that are needed.
2.) Call the install script, a good standard is install.sql below is an example.
spool install.lst

PROMPT Calling define_var.sql...

@@define_var.sql

/************************************************************************

* Creating OrderItems                                                       *

************************************************************************/

@cr_order_items_table.sql

spool off;

!grep 'ORA-' *.lst

!grep 'SP2-' *.lst

exit
There are some very important items to note in the small script above.  First is the define_var.sql file.  This file contains every variable used in the creation of database objects.  This allows the installation to be run on multiple databases without modifying the code.  

The next thing that should be noted is the “!grep 'ORA-' *.lst”  while this command looks very simple it’s also very powerful because it immediately shows any errors that might have occurred when the script was run and removes any chance that the operator will forget to review the log files.  This is a good example of how a small change can reduce a maintenance window.

3.) Call a post_validation script to verify objects and grants were created as expected.  Below is an example of a post validation script that is used to verify the Order Items Table was created correctly.  The post validation script uses a package called PkgValidation to assert that the actual results match the expected results.  The code is listed in Appendix C.  The Utplsql package could also be used for this purpose.
set serveroutput on size 1000000;

@@define_var.sql

spool post_validation.lst;

declare

v_actual 
varchar2(30);

begin

-- Verify TABLES 

select table_name into v_actual from user_tables

where table_name = 'ORDERITEMS'

  and tablespace_name = '&&DATA_TABLESPACE';

PkgValidation.assert('ORDERITEMS', v_actual,' Verify TABLE ORDERITEMS');

-- Verify CONSTRAINTS 

select count(*) into v_actual 

from user_constraints

where constraint_name = 'PK_ORDERITEMS'

  and constraint_type = 'P'

  and table_name      = 'ORDERITEMS';

PkgValidation.assert(1 ,v_actual,'Verify Primary Key PK_ORDERITEMS on ORDERITEMS');  

-- Verify INDEXES

select count(*) into v_actual

from user_indexes 

where index_name = 'PK_ORDERITEMS'

  and index_type = 'NORMAL'

  and table_name = 'ORDERITEMS'

  and uniqueness = 'UNIQUE'

  and tablespace_name = '&&INDEX_TABLESPACE';

PkgValidation.assert(1 ,v_actual,'Verify INDEX PK_ORDERITEMS on ORDERITEMS'); 

--

--

--  Print Validation Summary

--

dbms_output.put_line('=======================================================');

dbms_output.put_line('**Assertions passed : '||PkgValidation.v_assertions_passed);

dbms_output.put_line('**Assertions Failed : '||PkgValidation.v_assertions_failed);

dbms_output.put_line('**Total Assertions  : '||PkgValidation.v_total_assertions);

dbms_output.put_line('=======================================================');

end;

/

spool off;

Uninstalls

The uninstall procedure follows the same principals as the install procedure.  This script is very important in any environment because it allows the install procedure to be tested multiple times providing a higher probability that the install script will work in production.  In addition, if the install script fails in production you can quickly uninstall and re-install the release using these scripts.
Rollback

Rollback is a more difficult process.  This procedure is used to rollback the installation after transactions have occurred on the upgraded system.  To do this you must be able to modify the schema in such a way that old procedures and packages can still function with the modified schema. 
Implmenting a Daily Build Process

Implementing a daily build process is not as difficult as it seems.  If you are working with Oracle 10g there are many tools that can aid in this task.  The main benefit of the daily build process is that it enables you to immediately find code that will break the system.  Traditional approaches have many developers work on different pieces of the application and then perform builds when they want to provide code to QA or production.  During this build process, all of the components are integrated and flaws in the design or coding are found.  This is one of the worst times to find issues because at this point you are working against a quickly approaching deadline and have to quickly identify and resolve the issue.
Since you already have simple install and uninstall scripts with configuration files to drive the needed parameters and automated unit tests that can be run multiple times, the main components of a daily build process exist.

The only missing piece of the daily build is to write a script that will checkout the code from the configuration management repository such as Starteam, CVS or subversion and then run the install scripts and finally your unit tests.  This provides you with daily feedback on any problems with the code as early as possible.  

This build process also has a beneficial side effect.   Since you have been doing builds on a daily basis, the build for QA can just be one of your daily builds.   You no longer need to schedule time for a dedicated build process, integration tests and correcting integration errors that occurs during the build cycle.

One way to implement a repeatable daily build process is to use flashback database.  With this feature you can log the SCN of the database at the start of the build, then create all of the new objects and run your tests.  When complete you can rewind the database and start over again.  

An example of how this would work is:

 If flashdatabase has not been enabled 
· Set init.ora or spfile variables

db_recovery_file_dest = '/ora01/oradata/devdb/flash'

db_recovery_file_dest_size = 2g

db_flashback_retention_target = 2880


· Turn on flashback


SQL>Startup db in mount mode.


SQL>Startup mount;


SQL>alter database flashback on;


SQL> Alter database open;
1.) Capture the scn before the build process

select DBMS_FLASHBACK.GET_SYSTEM_CHANGE_NUMBER() from dual;

2.) 
-- Execute your build scripts and create any tables etc..

3.) Flashback DB.


SQL> Shutdown immediate


SQL> Startup mount;


SQL> FLASHBACK DATABASE TO SCN <value from step 1>;

SQL> alter database open resetlogs;

This procedure can be very useful when deploying code such as Oracle Advanced Queuing, Streams or Data Guard.
Creating a logging and monitoring framework 

Application developers are very familiar with logging frameworks, the most popular being log4j.  Log4j is a java logging package used in many java applications.  A logging package also exists for PL/SQL, that package is called LOG4PLSQL it can be found at http://log4plsql.sourceforge.net/. 
The importance of developing a logging framework is that it allows for easier debugging and centralized monitoring of your system.  In our implementation we created a table called eventlog with the following table structure.

CREATE TABLE EVENTLOG ( 

EVENTLOGID      NUMBER(9)                     NOT NULL,

SEVERITYID      NUMBER(2)   DEFAULT 1         NOT NULL,

LOCATION        VARCHAR2(100 BYTE)            NOT NULL,

ERRORMSG        VARCHAR2(4000 BYTE),

ADDITIONALINFO  VARCHAR2(4000 BYTE),

CREATEDDATE     DATE,

CREATEDBY       NUMBER(9)

) 

· EVENTLOGID is sequentially generated 

· SEVERITYID is the severity level of the message (1 - 4) 1 = info, 4 = error 

· LOCATION is the method/module/application from where the message originated 

· ERRORMSG is the type of error (info, reporterror, dberror) 

· ADDITIONALINFO is the additional information that the user would like to log 

· CREATEDDATE is the message created date 

· CREATEDBY is the user who created the message

Error messages would always be logged, Info messages would be logged only if the DEBUG flag is set to TRUE.
We then created a package named PKGEVENTLOG, this package insert records into the EVENTLOG table described above.
 PROCEDURE createeventlog (

    pseverityid       eventlog.severityid%TYPE,

    plocation         eventlog.LOCATION%TYPE,

    perrormsg         eventlog.errormsg%TYPE,

    padditionalinfo   eventlog.additionalinfo%TYPE,

    pcreateddate      eventlog.createddate%TYPE,

    pcreatedby        eventlog.createdby%TYPE

 );
Adding a debug mode to packages can greatly reduce the amount of time required to diagnose an error.  Many times developers will write debug statements and then remove them before sending the code to production.  By having a debug mode that is turned on only when needed the debugging statements can remain in the production source code and be available in case of emergency.  To ensure debugging messages are not slowing down your system a function to check if debug mode is enable needs to be written.  Below is the specification of the function. 

FUNCTION getdebugflag
  RETURN BOOLEAN;

Below are two examples of how the getdebugflag function would be used within a procedure.
Example 1 – log info message 

if (pkgeventlog. getdebugflag) then

  pkgeventlog.createeventlog(1,'PkgOrderItems.AddOrderItems','info','Get transid',sysdate,1);
end if;

Example 2 – log error message 

EXCEPTION WHEN OTHERS THEN

    vError:=substr(SQLERRM,1,100);

    pkgeventlog.createeventlog(4,'PkgOrderItems.AddOrderItems ','error',vError,sysdate,1);

    RAISE;

To add a  monitoring framework on top of this the only thing needed is to create a job to query the eventlog table for any errors in the table as well as keep track of the last error records that have been sent, so they are not sent multiple times.
Summary 

In conclusion, the techniques described in this document will help increase the quality of any database pl/sql code and deployment.  Creating unit tests can be done at any stage in a project.  If the project has just started, the tests can be written while the code is being developed.  For existing projects, tracing database sessions provides a great jump start.

Adding rigorous testing has amazing side effects to a project; developers have more confidence and are not distracted fixing defects. In addition, when requirements change late in the project, the existing tests provide empirical proof that the system still works properly.  Everything can be tested including packages, object creation scripts and reports.   It should be the responsibility of every engineer to prove his/her code is correct.    
Adding a monitoring and logging framework enables the rapid collection of diagnostic data and reduces the time required to identify the cause of an issue.  
The world is reliant on software systems like never before and the quality of those systems is of key importance to every information technology organization.  Every engineer should be trying to prove their software works properly.  This focus on testing increases the resilience of any software system and should be embraced by all engineers. 
Appendix - A  Create Order Items Table

-----------------begin------- define_var.sql----------------------------

DEFINE data_tablespace = DATA

DEFINE index_tablespace = INDEX

----------------end----------- define_var.sql------------------------

-----------------begin--------create_order_items_table.sql ------------------------------

@@define_var.sql

spool cr_order_items.lst

whenever sqlerror exit

Create table OrderItems

(Order_Id NUMBER(12) NOT NULL,

 Line_Item_Id NUMBER(3) NOT NULL,

 Product_Id NUMBER(6) NOT NULL,

 Unit_Price NUMBER(8,2),

 Quantity NUMBER(8)

)

TABLESPACE &&data_tablespace;

CREATE UNIQUE INDEX Pk_OrderItems ON OrderItems

(Order_Id, Line_Item_Id)

TABLESPACE &&index_tablespace;

ALTER TABLE OrderItems ADD (

  CONSTRAINT Pk_OrderItems

 PRIMARY KEY

 (Order_Id,Line_Item_Id));

GRANT SELECT ON  OrderItems TO select_role;

GRANT INSERT,SELECT ON  OrderItems TO read_insert_role;

spool off

-----------------end --------create_order_items_table.sql ------------------------------

Appendix - B  Create Order Items Package

CREATE OR REPLACE PACKAGE PkgOrderItems AS

  PROCEDURE Add_OrderItems(

  POrder_Id      IN  OrderItems.Order_Id%TYPE,

  PLine_Item_Id  IN  OrderItems.Line_Item_Id%TYPE,

  PProduct_Id    IN  OrderItems.Product_Id%TYPE,

  PUnit_Price    IN  OrderItems.Unit_Price%TYPE,

  PQuantity      IN  OrderItems.Quantity%TYPE);

END  PkgOrderItems;

/

CREATE OR REPLACE PACKAGE BODY PkgOrderItems AS

  PROCEDURE Add_OrderItems(

    POrder_Id      IN  OrderItems.Order_Id%TYPE,

    PLine_Item_Id  IN  OrderItems.Line_Item_Id%TYPE,

    PProduct_Id    IN  OrderItems.Product_Id%TYPE,

    PUnit_Price    IN  OrderItems.Unit_Price%TYPE,

    PQuantity      IN  OrderItems.Quantity%TYPE)

  IS

  BEGIN

   Insert into OrderItems(order_id,line_item_id,product_id,unit_price,quantity)

   values

   (pOrder_id, pLine_item_id,pProduct_id,Punit_price,pQuantity);

   END Add_OrderItems;

END PkgOrderItems;

/

Appendix - C  PkgValidation.sql
CREATE OR REPLACE PACKAGE PkgValidation AS

  v_total_assertions
NUMBER := 0;

  v_assertions_passed   NUMBER := 0;

  v_assertions_failed   NUMBER := 0;

  PROCEDURE Assert ( 
 

  

      p_assertion in varchar,

  

      p_actual in varchar,



      p_description in varchar2 );

END  PkgValidation;

/

--

CREATE OR REPLACE PACKAGE BODY PkgValidation AS


Procedure Assert ( 





p_assertion in varchar,




p_actual in varchar,




p_description in varchar2 ) IS


ERROR 




VARCHAR2(200);


ASSERTION_FAILED


EXCEPTION;


BEGIN


   ERROR := 'ORA-99999: Assertion Failed! '||p_description;


   PkgValidation.v_total_assertions  := PkgValidation.v_total_assertions + 1;


   if p_assertion != p_actual then



RAISE ASSERTION_FAILED;


   end if;


   dbms_output.put_line('PASSED: Assertion '||p_description||chr(10));


   dbms_output.new_line();


   PkgValidation.v_assertions_passed := PkgValidation.v_assertions_passed + 1;


exception


   when ASSERTION_FAILED then


      dbms_output.put_line(ERROR);


      dbms_output.put_line('ASSERTION :'||p_assertion||' ACTUAL :'||p_actual||chr(10));


      dbms_output.new_line();


      PkgValidation.v_assertions_failed := PkgValidation.v_assertions_failed + 1;


END Assert;

END PkgValidation;


/

